



Séries de Fourier

Dada uma função $f(x)$ no intervalo $-L < x < L$, este procedimento calcula a_n e b_n se $n > 0$ e calcula $a_0/2$ se $n=0$. Há um plot que ilustra a aproximação das séries de Fourier da função. Execute esta worksheet e faça os exemplos.

> **restart:**

Dado $f(x)$ em $-L < x < L$, calculamos a_n .

```
> fourier_cos_coeff:=proc(f,L,n::integer)
> local x,a;
> if n=0 then a:=int(f(x),x=(-L)..L)/(2*L); fi;
> if n>0 then a:=int(f(x)*cos(n*Pi*x/L),x=(-L)..L)/L; fi;
> RETURN(a);
> end;
```

Dado $f(x)$ em $-L < x < L$, calculamos b_n se $n > 0$.

```
> fourier_sin_coeff:=proc(f,L,n::integer)
> local x,b;
> b:=int(f(x)*sin(n*Pi*x/L),x=(-L)..L)/L;
> RETURN(b);
> end;
```

Exemplo 1


```
> with(linalg): f:=x->piecewise(x<0,1,x>0,x);
```

Warning, new definition for norm

Warning, new definition for trace

$$f := x \rightarrow \text{piecewise}(x < 0, 1, 0 < x, x)$$

```
> S2:=fourier_cos_coeff(f,5,0)+add(innerprod(vector(2,[fourier_cos_coeff(f,5,n),fourier_sin_coeff(f,5,n)]), vector(2, [cos(n*Pi*x/5),sin(n*Pi*x/5)])),n=1..20);  
> plot([f(x),S2],x=-5..5);
```




>

Exemplo 2

```
> g:=x->x^2;
```

$$g := x \rightarrow x^2$$

```
> S2:=fourier_cos_coeff(g,1,0)+add(innerprod(vector(2,[fourier_cos_coeff(g,1,n),fourier_sin_coeff(g,1,n)]), vector(2, [cos(n*Pi*x/1),sin(n*Pi*x/1)])),n=1..5);  
> plot([g(x),S2],x=-1..1);
```




Observação: como plotar extensões periódicas de funções dadas. Defina pex e troca depois x pela função que você quer. Veja o exemplo.


```
> pex:=(x,p)->-p+ x+p-floor((x+p)/(2*p))*2*p;
```

$$pex := (x, p) \rightarrow x - 2 \text{ floor}\left(\frac{1}{2} \frac{x+p}{p}\right)p$$

```
> plot(subs(x=pex(x,Pi),exp(x)),x=-10..10);
```



```
> plot(subs(x=pex(x,Pi),x^2),x=-10..10);
```



```
>
```